package com.waitingsnow.core.service
{

import com.waitingsnow.core.debug.Logger;

import flash.net.SharedObject;

/**
 * 管理客户端ShareObject 指定属性避免使用关键字比如name等 会造成过期后无法删除
 * @author waitingsnow
 * $Id: CookieService.as 2 2012-05-17 10:10:03Z cgslendy@gmail.com $
 */
public class CookieService
{
	//==========================================================================
	//  Class variables
	//==========================================================================
	static private const EXPIRES_NAME:String = "expires";
	
	//==========================================================================
	//  Constructor
	//==========================================================================
	/**
	 * Constructs a new <code>CookieService</code> instance.
	 * @param cookieName
	 */
	public function CookieService(cookieName:String)
	{
		_cookieName = cookieName;
		init();
	}
	
	//==========================================================================
	//  Variables
	//==========================================================================
	private var so:SharedObject;
	
	//==========================================================================
	//  Properties
	//==========================================================================
	private var _cookieName:String;
	/**
	 * cookie名字
	 */
	public function get cookieName():String
	{
		return _cookieName;
	}
	
	private var _expires:int;
	/**
	 * 过期时间单位s 0表示不过期 过期后so所有属性会被删除
	 */
	public function get expires():int
	{
		return _expires;
	}
	/**
	 * @private
	 */
	public function set expires(value:int):void
	{
		_expires = value;
		setProps(EXPIRES_NAME, (new Date()).time + _expires*1000);
	}

	//==========================================================================
	//  Methods
	//==========================================================================
	/**
	 * 检查是否存在该文件 并检查一次文件是否过期
	 * @return 
	 * 
	 */	
	public function init():void
	{
		try
		{
			so = SharedObject.getLocal(_cookieName);
			var soObj:Object = so.data;
			if (soObj.hasOwnProperty(EXPIRES_NAME) && 
				(soObj[EXPIRES_NAME] <= (new Date()).time))
			{
				for (var key:* in so.data)
				{
					delete so.data[key];
				}
				so.flush();
			}
		}
		catch (error:Error)
		{
			Logger.error(_cookieName, "SharedObject被禁止访问,错误代码： ", error.toString());
		}
	}
	
	/**
	 * 得到指定属性值
	 * @param name
	 * @return 
	 * 
	 */	
	public function getProps(name:*):*
	{
		return contains(name) ? so.data[name] : null;
	}
	
	/**
	 * 设置指定属性值
	 * @param name
	 * @param value
	 * 
	 */	
	public function setProps(name:*, value:*):void
	{
		if (!so)
		{
			return;
		}
		try
		{
			so.data[name] = value;
			so.flush();
		}
		catch (error:Error)
		{
			Logger.error("SharedObject写入: ", name, value, "失败。错误代码： ", error.toString());
		}
	}
	
	public function getObject():Object
	{
		return so && so.data;
	}
	
	/**
	 * 删除指定属性
	 * @param name
	 * 
	 */	
	public function delProps(name:*):void
	{
		if (contains(name))
		{
			try
			{
				delete so.data[name];
				so.flush();
			}
			catch (error:Error)
			{
				Logger.error("SharedObject删除：", name, "失败。错误代码：", error.toString());
			}
		}
	}
	
	/**
	 * 删除所有属性 并删除本身
	 */	
	public function removeAllProps():void
	{
		so && so.clear();
	}
	
	/**
	 * 释放
	 */	
	public function dispose():void
	{
		so = null;
	}
	
	/**
	 * 是否包含该键值
	 * @param name
	 * @return 
	 * 
	 */	
	private function contains(name:*):Boolean
	{
		return (so && so.data.hasOwnProperty(name)) ? true : false;
	}
}
}